```
<< HolonomicFunctions.m
```

HolonomicFunctions package by Christoph Koutschan, RISC−Linz, Version 1.4 (10.11.2010)
⟶ Type ?HolonomicFunctions for help

```
(* Generalizations of Theorem 2.9 to Jacobi polynomials *)

jacann = Factor[Annihilator[JacobiP[n, a, b, x], {S[n], S[a], S[b], Der[x]}]]
```

$$\{(1+a+b+n)\,S_b + (1-x)\,D_x + (-1-a-b-n),\ (1+a+b+n)\,S_a + (-1-x)\,D_x + (-1-a-b-n),$$
$$2\,(1+n)\,(1+a+b+n)\,S_n - (2+a+b+2n)\,(-1+x)\,(1+x)\,D_x - (1+a+b+n)\,(a-b+2x+a\,x+b\,x+2\,n\,x),$$
$$(-1+x)\,(1+x)\,D_x^2 + (a-b+2x+a\,x+b\,x)\,D_x - n\,(1+a+b+n)\}$$

```
Factor[FindRelation[jacann, Support → {S[n], S[n]^2 Der[x], Der[x]}]]

FindRelation[jacann, Support → {S[a] S[b], S[a] S[b] S[n], S[a] S[b] S[n]^2, S[n]^2}]
```

$$\{-(3+a+b+n)\,(a-b+4x+a\,x+b\,x+2\,n\,x)\,S_n S_a S_b + 2\,(2+n)\,(4+a+b+2n)\,S_n^2 +$$
$$2\,(2+a+n)\,(2+b+n)\,S_a S_b,\ 2\,(2+n)\,(4+a+b+n)\,(4+a+b+2n)\,S_n^2 S_a S_b - (5+a+b+2n)$$
$$\left(2a+a^2-2b-b^2+24x+10\,a\,x+a^2x+10\,b\,x+2\,a\,b\,x+b^2x+20\,n\,x+4\,a\,n\,x+4\,b\,n\,x+4\,n^2x\right)$$
$$S_n S_a S_b + 2\,(2+a+n)\,(2+b+n)\,(6+a+b+2n)\,S_a S_b\}$$

```
(* Linearization coefficients for products of Legendre polynomials *)

time =
 Timing[ann = Factor[Annihilator[LegendreP[k, x] LegendreP[m, x] LegendreP[n, x] (2 k + 1) / 2,
       {S[k], S[m], S[n], Der[x]}]];][[1]]
```

1.73611

```
time = Timing[conn = Factor[CreativeTelescoping[ann, Der[x], {S[k], S[m], S[n]}]];][[1]]
```

1644.21

```
(* principal part *)

conn[[1]]
```

$$\{(1+k+m-n)\,(k-m+n)\,S_m - (k+m-n)\,(1+k-m+n)\,S_n,$$
$$(1+2k)\,(k-m-n)\,(1+k+m-n)\,S_k - (3+2k)\,(-1+k-m-n)\,(k+m-n)\,S_n,$$
$$(-2+k-m-n)\,(-1+k+m-n)\,(2+k-m+n)\,(3+k+m+n)\,S_n^2 -$$
$$(-1+k-m-n)\,(k+m-n)\,(1+k-m+n)\,(2+k+m+n)\}$$

```
(* delta part: one needs to verify that this vanishes indeed (factors: (1-x)(1+x)) *)

conn[[2]]

(* determine a recurrence solely in shifts in k *)

krec = Factor[FindRelation[conn[[1]], Support → {1, S[k], S[k]^2}]]
```

$$\{(1+2k)\,(1+k-m-n)\,(2+k+m-n)\,(2+k-m+n)\,(3+k+m+n)\,S_k^2 -$$
$$(5+2k)\,(k-m-n)\,(1+k+m-n)\,(1+k-m+n)\,(2+k+m+n)\}$$

```
krec1 = ApplyOreOperator[krec, a[k]]
```

$$\{-(5+2k)\,(k-m-n)\,(1+k+m-n)\,(1+k-m+n)\,(2+k+m+n)\,a[k] +$$
$$(1+2k)\,(1+k-m-n)\,(2+k+m-n)\,(2+k-m+n)\,(3+k+m+n)\,a[2+k]\}$$

```
FullSimplify[RSolve[First[krec1] == 0, a[k], k]]
```

$$\left\{\left\{a[k] \to -\left((1+2k)(m-n)(1+m+n)\left(C[1]+(-1)^k C[2]\right)\text{Cos}\left[\frac{1}{2}(m-n)\pi\right]\text{Cos}\left[\frac{1}{2}(-1+m+n)\pi\right]\text{Gamma}\left[\frac{1}{2}(k-m-n)\right]\text{Gamma}\left[\frac{1}{2}(1+k+m-n)\right]\text{Gamma}\left[\frac{1}{2}(1+k-m+n)\right]\text{Gamma}\left[\frac{1}{2}(2+k+m+n)\right]\right)\Big/\left(2\,\text{Gamma}\left[\frac{1}{2}(1+k-m-n)\right]\text{Gamma}\left[\frac{1}{2}(2+k+m-n)\right]\text{Gamma}\left[\frac{1}{2}(2+k-m+n)\right]\text{Gamma}\left[\frac{1}{2}(3+k+m+n)\right](\text{Sin}[m\pi]-\text{Sin}[n\pi])\right)\right\}\right\}$$

```
(* product recurrence *)
```

```
ann = Factor[Annihilator[JacobiP[i, 2, 0, x] JacobiP[j, 1, 1, x], {S[i], S[j]}]]
```

$$\left\{(2+j)(4+j)S_j^2-(3+j)(5+2j)\,x\,S_j+(2+j)(3+j),\ (2+i)^2(4+i)S_i^2-(5+2i)\left(1+6x+5ix+i^2x\right)S_i+(1+i)(3+i)^2\right\}$$

```
rec = FindRelation[ann, Eliminate → {x}] // Factor
```

$$\left\{(2+i)^2(4+i)(3+j)(5+2j)S_i^2S_j-(2+i)(3+i)(5+2i)(2+j)(4+j)S_iS_j^2-(5+2i)(3+j)(5+2j)S_iS_j-(2+i)(3+i)(5+2i)(2+j)(3+j)S_i+(1+i)(3+i)^2(3+j)(5+2j)S_j\right\}$$

```
Support[rec]
```

$$\left\{\left\{S_i^2S_j,\ S_iS_j^2,\ S_iS_j,\ S_i,\ S_j\right\}\right\}$$

```
(* extract the recurrence from the annihilator *)
```

```
rec1 = ApplyOreOperator[rec, A[i, j]]
```

$$\{(1+i)(3+i)^2(3+j)(5+2j)A[i,1+j]-(2+i)(3+i)(5+2i)(2+j)(3+j)A[1+i,j]-(5+2i)(3+j)(5+2j)A[1+i,1+j]-(2+i)(3+i)(5+2i)(2+j)(4+j)A[1+i,2+j]+(2+i)^2(4+i)(3+j)(5+2j)A[2+i,1+j]\}$$

```
(* graphical interpretation of the support *)
```

```
Show[Graphics[{PointSize[0.1],
   Point[#] & /@ (Cases[rec1, A[_, _], Infinity] /. A[i + a_., j + b_.] → {a, b})}]]
```



```
Collect[Solve[(rec1 /. {i → i - 1, j → j - 2}) == 0, A[i, j]][[1, 1, 2]], A[_, _], Factor]
```

$$\frac{i\,(2+i)\,(1+j)\,(1+2j)\,A[-1+i,\,-1+j]}{(1+i)\,(3+2i)\,j\,(2+j)} - \frac{(1+j)\,A[i,\,-2+j]}{2+j} - \frac{(1+j)\,(1+2j)\,A[i,\,-1+j]}{(1+i)\,(2+i)\,j\,(2+j)} + \frac{(1+i)\,(3+i)\,(1+j)\,(1+2j)\,A[1+i,\,-1+j]}{(2+i)\,(3+2i)\,j\,(2+j)}$$

```
(* compute the matrix entries recursively *)

ClearAll[A];
A[-1, j_Integer] := 0;
A[i_Integer, -1] := 0;
A[i_Integer, 0] := 2;
A[i_Integer, j_Integer] :=
  A[i, j] = i (2 + i) (1 + j) (1 + 2 j) A[-1 + i, -1 + j] / ((1 + i) (3 + 2 i) j (2 + j)) - (1 + j) A[i, -2 + j] / (2 + j) -
    (1 + j) (1 + 2 j) A[i, -1 + j] / ((1 + i) (2 + i) j (2 + j)) + (1 + i) (3 + i) (1 + j) (1 + 2 j) A[1 + i, -1 + j] / ((2 + i) (3 + 2 i) j (2 + j));

Timing[mat = Table[A[i, j], {i, 0, 50}, {j, 0, 50}];]
```

{0.224015, Null}

```
Table[A[i, j], {i, 0, 15}, {j, 0, 15}] // MatrixForm
```

2	0	1	0	2/3	0	1/2	0	2/5	0	1/3	0	2/7	0	1/4	0
2	8/3	1	8/5	2/3	8/7	1/2	8/9	2/5	8/11	1/3	8/13	2/7	8/15	1/4	8/17
2	10/3	7/2	2	7/3	10/7	7/4	10/9	7/5	10/11	7/6	10/13	1	2/3	7/8	10/17
2	18/5	9/2	22/5	3	22/7	9/4	22/9	9/5	2	3/2	22/13	9/7	22/15	9/8	22/17
2	56/15	5	28/5	16/3	4	4	28/9	16/5	28/11	8/3	28/13	16/7	28/15	2	28/17
2	80/21	37/7	44/7	20/3	44/7	5	44/9	4	4	10/3	44/13	20/7	44/15	5/2	44/17
2	27/7	153/28	47/7	15/2	54/7	29/4	6	29/5	54/11	29/6	54/13	29/7	18/5	29/8	54/17
2	35/9	67/12	7	145/18	26/3	35/4	74/9	7	74/11	35/6	74/13	5	74/15	35/8	74/17
2	176/45	17/3	36/5	76/9	28/3	49/5	88/9	46/5	8	23/3	88/13	46/7	88/15	23/4	88/17
2	216/55	63/11	404/55	96/11	108/11	581/55	120/11	54/5	112/11	9	112/13	54/7	112/15	27/4	112/17
2	130/33	127/22	82/11	295/33	112/11	245/22	388/33	12	130/11	67/6	10	67/7	26/3	67/8	130/17
2	154/39	151/26	98/13	355/39	136/13	301/26	484/39	168/13	170/13	77/6	158/13	11	158/15	77/8	158/17
2	360/91	531/91	692/91	120/13	972/91	155/13	168/13	1242/91	1280/91	99/7	180/13	92/7	12	23/2	180/17
2	416/105	41/7	268/35	28/3	76/7	61/5	40/3	498/35	104/7	319/21	76/5	104/7	212/15	13	212/17
2	119/30	47/8	77/10	113/12	11	497/40	41/3	147/10	31/2	385/24	163/10	65/4	238/15	121/8	14
2	135/34	801/136	263/34	645/68	189/17	1715/136	237/17	513/34	545/34	2277/136	585/34	1183/68	294/17	135/8	274/17